
1.1 Change 1

	Testcase name
	7.2.3.21

	Protocol
	RLC

	Function Name
	f_TC_7_2_3_21_EUTRA

	Reason for change
	According to 36.508 the PDCP status report should be disable for all the RLC and MAC handover test case.

	Summary of change
	Disable the PDCU Status report.

	Source of change
	File : RLC_AM_Testcases.ttcn


Before:

	function f_TC_7_2_3_21_EUTRA() runs on EUTRA_PTC

{ // @sic R5s100256 sic@ @sic RP-101255 sic@

  var RLC_SS_State v_RLC_Rec;

  …………………
  …………………
  …………………
  fl_EUTRA_GenericRbEst_WithRLC_AM_Params (eutra_Cell1,

                                           {cs_DRB_ToAddMod_RLC_AM (tsc_DRB1,

                                                                    cds_TX_AM_RLC_t_PollRetransmit (ms150),

                                                                    cds_RX_AM_RLC_t_Reord (ms150))});
  …………………
  …………………
  …………………
  // postamble

  f_EUTRA_Postamble (eutra_Cell1, E3_TESTMODE);

} // f_TC_7_2_3_21_EUTRA


After:

	function f_TC_7_2_3_21_EUTRA() runs on EUTRA_PTC

{ // @sic R5s100256 sic@ @sic RP-101255 sic@

  var RLC_SS_State v_RLC_Rec;

  …………………
  …………………
  …………………
  f_EUTRA_RbEst_Def_DisableBSR_PHR_72321(eutra_Cell1, 

                                            {cs_DRB_ToAddMod_RLC_AM_72321 (tsc_DRB1,

                                                                           cds_TX_AM_RLC_t_PollRetransmit (ms150),

                                                                           cds_RX_AM_RLC_Common (ms150, ms60))},

                                            false 

                                            );
  …………………
  …………………
  …………………
  // postamble

  f_EUTRA_Postamble (eutra_Cell1, E3_TESTMODE);

} // f_TC_7_2_3_21_EUTRA


New Functions: 

	  function f_EUTRA_RbEst_Def_DisableBSR_PHR_72321(CellId_Type p_CellId,

                                                     template (value) DRB_ToAddModList p_DRB_ToAddModList,

                                                     boolean p_PDCP_StatusReportRequired := true) runs on EUTRA_PTC

  {

    fl_EUTRA_GenericRbEst_DisableBSR_PHR_72321(p_CellId, 0, 0, omit, omit, p_DRB_ToAddModList, p_PDCP_StatusReportRequired);

  };


	  function fl_EUTRA_GenericRbEst_DisableBSR_PHR_72321( CellId_Type p_CellId,

                                                       integer p_NoOfAddDRBs_AM,

                                                       integer p_NoOfAddDRBs_UM,

                                                       template (omit) EPS_QoSList_Type p_QoS,

                                                       template (omit) Tft_List_Type p_Tft,

                                                       template (value) DRB_ToAddModList p_DRB_ToAddModList,

                                                       boolean p_PDCP_StatusReportRequired := true) runs on EUTRA_PTC

  {

    var template (value) MAC_MainConfig_Type v_MAC_MainConfig ;

    var template (omit) PhysicalConfigDedicated v_PhysicalConfigDedicated := omit;  // => apply default setting acc. 36.508 cl. 4.8.2.1.6 */

    if (p_NoOfAddDRBs_UM > 0) {

      v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxS_DisableBSR_PHR; // short DRX cycle acc. to note in 36.508 table 4.6.3-16

    } else {

      v_MAC_MainConfig := cs_508_MAC_MainConfig_Explicit_RBC_DrxL_DisableBSR_PHR; // long DRX cycle acc. to note in 36.508 table 4.6.3-16

    };

    f_EUTRA_GenericRbEst_Common(p_CellId,

                                p_NoOfAddDRBs_AM,

                                p_NoOfAddDRBs_UM,

                                p_DRB_ToAddModList,

                                v_MAC_MainConfig,

                                v_PhysicalConfigDedicated,

                                p_QoS,

                                p_Tft,   

                                p_PDCP_StatusReportRequired);

  };



1.2 Change 2

	Testcase name
	7.2.3.21

	Protocol
	RLC

	Function Name
	fl_EUTRA_7_2_3_21_TestBody

	Reason for change
	1)  According to 36.523-3, px_DelayBeforeIntraCellHO paramter should be used to configure the delay to allow the configuration for the handover.

2) At step 2, SS is expecting PDU from UE and SS is configuring the DRB data and handover message in advance so while configuring the handover message SS should not wait the handover complete message. 

	Summary of change
	1) Use px_DelayBeforeIntraCellHO paramter is to configure delay to allow the configuration for the handover.

2) Expect the handover complete message after receiving he DRB data from UE at step 2.

	Source of change
	File : RLC_AM_Testcases.ttcn


Before:

	function fl_EUTRA_7_2_3_21_TestBody(inout RLC_SS_State p_RLC_Rec)

  runs on EUTRA_PTC

{ //@sic R5-104723 sic@ @sic RP-101255 sic@

  var integer i;

  var RLC_DataField_Type v_RLC_Data1;

  var integer v_PDCP_SDU_size := 38;

  var template (value) PDCP_DataPdu_LongSN_Type v_PDCP_DataPDU;

  var RLC_DataField_Type v_SDU3_Received;

  var RLC_DataField_Type v_RLC_Data;

  var SubFrameTiming_Type v_Timing;

  var integer v_SubframeOffset;

  var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode;

  timer t_WatchDog := 5.0;

  // determine RAT to make the timing correct later on

  v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);

  // PDCP SDUs of this size will be built, appended to a PDCP header,

  // and then used as RLC SDUs

  // generate 3 RLC SDUs = PDCP PDUs

  for (i := 0; i < 3; i:= i + 1) {

    p_RLC_Rec.TxDataSize := v_PDCP_SDU_size;

    f_GetTxAM_PRBS (p_RLC_Rec);  // sets p_RLC_Rec.RLC_Data

    v_PDCP_DataPDU := cs_PDCP_DataPdu_LongSN(int2bit(i, tsc_PDCP_SN_Size_12), p_RLC_Rec.RLC_Data);

    p_RLC_Rec.RLC_DataList[i] := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(v_PDCP_DataPDU));

    p_RLC_Rec.Next_PDCP_TX_SN := (p_RLC_Rec.Next_PDCP_TX_SN + 1) mod (tsc_Maximum_PDCP_SN_12 + 1);

  }

  v_RLC_Data := substr(p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3],2,(v_PDCP_SDU_size));

  v_SDU3_Received := f_PDCP_PDU_Encvalue (cs_PDCP_DataLongSN(1,v_RLC_Data));

  //@siclog "Step 0" siclog@

  // The SS ignores scheduling requests and does not allocate any uplink grant based on that.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  //@siclog "Step 1" siclog@

  // SS creates 3 RLC SDUs of size 40 bytes segmented into two AMD PDUs each.

  // SS transmits AMD PDU#1 (SN=0), AMD PDU#2 (SN=1) and AMD PDU#4 (SN=3).

  v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1,100);
  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 0, 20);

  f_TxAMD_OnePDU (p_RLC_Rec,

                  cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                  tsc_P_NoPoll,

                  tsc_FI_StartOfSDU,

                  v_RLC_Data1);

  v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 1);

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 20, 20);

  f_TxAMD_OnePDU (p_RLC_Rec,

                  cs_TimingInfo(v_Timing.SFN.Number, v_SubframeOffset),

                  tsc_P_NoPoll,

                  tsc_FI_EndOfSDU,

                  v_RLC_Data1);

  p_RLC_Rec.AM_VTS := 3;

  v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 2);

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 20, 20);

  f_TxAMD_OnePDU (p_RLC_Rec,

                  cs_TimingInfo(v_Timing.SFN.Number, v_SubframeOffset),

                  tsc_P_NoPoll,

                  tsc_FI_EndOfSDU,

                  v_RLC_Data1);

  //@siclog "Step 1A" siclog@

  // 60 ms after step 1 the SS allocates 1 UL grant of default size

  v_Timing := f_EUTRA_TimingInfoAdd(v_Timing, 60);

  f_EUTRA_OneULGrantTransmission(eutra_Cell1,

                                 cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                 9, 25);

  //@siclog "Step 4" siclog@

  // 90 ms after step 1 SS performs a RRC Connection Reconfiguration procedure including the

  // mobilityControlInfo IE triggering RLC-reestablishment.

  v_Timing := f_EUTRA_TimingInfoAdd(v_Timing, 30); // i.e. 30 ms after step 1A

  f_EUTRA_508RRC_IntraLTE_HO_IntraCell (eutra_Cell1,  // @sic R5s100430 change 2; R5s100524 sic@

                                        cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                        {cs_DRB1_ConfigAM_TM_Mode});
  //@siclog "Step 2" siclog@

  // The UE returns RLC SDU#1.

  f_RxAMD_SDU(p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU1);

  //@siclog "Step 3" siclog@

  // SS does not acknowledge the reception of RLC SDU#1.

  p_RLC_Rec.AM_VRR := 0;

  //@siclog "Step 4AA" siclog@

  // The SS starts the UL default grant transmissions.

  f_EUTRA_StartDefULGrantTransmission(eutra_Cell1, cs_TimingInfo_Now);

  //@siclog "Step 4A" siclog@

  // The UE retransmits RLC SDU #1.

  f_RxAMD_SDU(p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU1);

  //@siclog "Step 4B" siclog@

  // Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  //@siclog "Step 5" siclog@

  // SS transmits AMD PDU#5 with SN=0 and the P field set to "1"

  p_RLC_Rec.AM_VTS := 0;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 0, 20);

  f_TxAMD_OnePDU (p_RLC_Rec,

                  cs_TimingInfo_Now,

                  tsc_P_Poll,

                  tsc_FI_StartOfSDU,

                  v_RLC_Data1);

  //@siclog "Step 6" siclog@

  // Check: Does the UE transmit a RLC STATUS PDU?

  p_RLC_Rec.AM_VTA := 1;

  f_RxSTATUS_PDU_ACK(p_RLC_Rec);

  //@siclog "Step 7" siclog@

  // SS transmits AMD PDU#6 with SN=Receiving_AM_Window_Size+2

  p_RLC_Rec.AM_VTS := tsc_AM_Window_Size + 2;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 20, 20);

  f_TxAMD_OnePDU (p_RLC_Rec,

                  cs_TimingInfo_Now,

                  tsc_P_NoPoll,

                  tsc_FI_EndOfSDU,

                  v_RLC_Data1);

  //@siclog "Step 8" siclog@

  // Check: Does UE return RLC SDU#3 within 1s?

  p_RLC_Rec.AM_VRR := 0;

  t_WatchDog.start(1.0);

  alt {

    [] DRB.receive (car_DRB1_RLCAmDataInd (eutra_Cell1,

                                           cr_AMD_PDUoneSDU(p_RLC_Rec.AM_VRR, tsc_P_NoPoll, {p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3]} )))

      {

        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 8");

      }

    [] t_WatchDog.timeout

      {

        // UE does not return RLC SDU#3 within 1s

      }

  }; // alt

  //@siclog "Step 9" siclog@

  // SS transmits AMD PDU#6 with SN=1

  p_RLC_Rec.AM_VTS := 1;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 20, 20);

  f_TxAMD_OnePDU (p_RLC_Rec,

                  cs_TimingInfo_Now,

                  tsc_P_NoPoll,

                  tsc_FI_EndOfSDU,

                  v_RLC_Data1);

  //@siclog "Step 10" siclog@

  // Check: Does UE return RLC SDU#3 with its first AMD PDU set to SN=1?

  p_RLC_Rec.AM_VRR := 1;

  DRB.receive (car_DRB1_RLCAmDataInd (eutra_Cell1,

                                      cr_AMD_PDUoneSDU(p_RLC_Rec.AM_VRR,

                                                       tsc_P_Poll,

                                                       {v_SDU3_Received}))) ;

  p_RLC_Rec.AM_VRR := (p_RLC_Rec.AM_VRR + 1) mod 1024;

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 10");

  //@siclog "Step 10A" siclog@

  // Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

} // fl_EUTRA_7_2_3_21_TestBody


After:

	function fl_EUTRA_7_2_3_21_TestBody(inout RLC_SS_State p_RLC_Rec)

  runs on EUTRA_PTC

{ //@sic R5-104723 sic@ @sic RP-101255 sic@

  var integer i;

  var RLC_DataField_Type v_RLC_Data1;

  var integer v_PDCP_SDU_size := 38;

  var template (value) PDCP_DataPdu_LongSN_Type v_PDCP_DataPDU;

  var RLC_DataField_Type v_SDU3_Received;

  var RLC_DataField_Type v_RLC_Data;

  var SubFrameTiming_Type v_Timing;

  var integer v_SubframeOffset;

  var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode;

  timer t_WatchDog := 5.0;

  // determine RAT to make the timing correct later on

  v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (eutra_Cell1);

  // PDCP SDUs of this size will be built, appended to a PDCP header,

  // and then used as RLC SDUs

  // generate 3 RLC SDUs = PDCP PDUs

  for (i := 0; i < 3; i:= i + 1) {

    p_RLC_Rec.TxDataSize := v_PDCP_SDU_size;

    f_GetTxAM_PRBS (p_RLC_Rec);  // sets p_RLC_Rec.RLC_Data

    v_PDCP_DataPDU := cs_PDCP_DataPdu_LongSN(int2bit(i, tsc_PDCP_SN_Size_12), p_RLC_Rec.RLC_Data);

    p_RLC_Rec.RLC_DataList[i] := f_PDCP_PDU_Encvalue (cs_PDCP_PDU_LongSN(v_PDCP_DataPDU));

    p_RLC_Rec.Next_PDCP_TX_SN := (p_RLC_Rec.Next_PDCP_TX_SN + 1) mod (tsc_Maximum_PDCP_SN_12 + 1);

  }

  v_RLC_Data := substr(p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3],2,(v_PDCP_SDU_size));

  v_SDU3_Received := f_PDCP_PDU_Encvalue (cs_PDCP_DataLongSN(1,v_RLC_Data));

  //@siclog "Step 0" siclog@

  // The SS ignores scheduling requests and does not allocate any uplink grant based on that.

  f_EUTRA_StopULGrantTransmission_PUCCH_Sync ( eutra_Cell1, cs_TimingInfo_Now );

  //@siclog "Step 1" siclog@

  // SS creates 3 RLC SDUs of size 40 bytes segmented into two AMD PDUs each.

  // SS transmits AMD PDU#1 (SN=0), AMD PDU#2 (SN=1) and AMD PDU#4 (SN=3).

  v_Timing := f_EUTRA_GetNextSendOccasion(eutra_Cell1, px_DelayBeforeIntraCellHO); // NOTE 1
  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 0, 20);

  f_TxAMD_OnePDU (p_RLC_Rec,

                  cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                  tsc_P_NoPoll,

                  tsc_FI_StartOfSDU,

                  v_RLC_Data1);

  v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 1);

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU1], 20, 20);

  f_TxAMD_OnePDU (p_RLC_Rec,

                  cs_TimingInfo(v_Timing.SFN.Number, v_SubframeOffset),

                  tsc_P_NoPoll,

                  tsc_FI_EndOfSDU,

                  v_RLC_Data1);

  p_RLC_Rec.AM_VTS := 3;

  v_SubframeOffset := f_EUTRA_TimingInfoSubframeOffset(v_EUTRA_FDD_TDD_Mode, 2);

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU2], 20, 20);

  f_TxAMD_OnePDU (p_RLC_Rec,

                  cs_TimingInfo(v_Timing.SFN.Number, v_SubframeOffset),

                  tsc_P_NoPoll,

                  tsc_FI_EndOfSDU,

                  v_RLC_Data1);

  //@siclog "Step 1A" siclog@

  // 60 ms after step 1 the SS allocates 1 UL grant of default size

  v_Timing := f_EUTRA_TimingInfoAdd(v_Timing, 60);

  f_EUTRA_OneULGrantTransmission(eutra_Cell1,

                                 cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                 9, 25);

  //@siclog "Step 4" siclog@

  // 90 ms after step 1 SS performs a RRC Connection Reconfiguration procedure including the

  // mobilityControlInfo IE triggering RLC-reestablishment.

  v_Timing := f_EUTRA_TimingInfoAdd(v_Timing, 30); // i.e. 30 ms after step 1A

  f_EUTRA_508RRC_IntraLTE_HO_IntraCell_72321 (eutra_Cell1,  // @sic R5s100430 change 2; R5s100524 sic@

                                              cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                              {cs_DRB1_ConfigAM_TM_Mode}); // NOTE 2
  //@siclog "Step 2" siclog@

  // The UE returns RLC SDU#1.

  f_RxAMD_SDU(p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU1);

  SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));       // @sic R5s100430  Change 2 sic@

  // configure RACH procedure for target cell back to default    @sic R5s100520 change 11 sic@

  f_EUTRA_SS_ConfigRachProcedure (eutra_Cell1,

                                  omit,

                                  cs_RachProcedureConfig_Def(omit, f_EUTRA_CellInfo_GetRAR_TA(eutra_Cell1), f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1)));

  f_EUTRA_StartDefULGrantTransmission(eutra_Cell1,cs_TimingInfo_Now); // NOTE 2
  //@siclog "Step 3" siclog@

  // SS does not acknowledge the reception of RLC SDU#1.

  p_RLC_Rec.AM_VRR := 0;

  //@siclog "Step 4AA" siclog@

  // The SS starts the UL default grant transmissions.

  f_EUTRA_StartDefULGrantTransmission(eutra_Cell1, cs_TimingInfo_Now);

  //@siclog "Step 4A" siclog@

  // The UE retransmits RLC SDU #1.

  f_RxAMD_SDU(p_RLC_Rec, tsc_P_Poll, tsc_RLC_SDU1);

  //@siclog "Step 4B" siclog@

  // Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

  //@siclog "Step 5" siclog@

  // SS transmits AMD PDU#5 with SN=0 and the P field set to "1"

  p_RLC_Rec.AM_VTS := 0;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 0, 20);

  f_TxAMD_OnePDU (p_RLC_Rec,

                  cs_TimingInfo_Now,

                  tsc_P_Poll,

                  tsc_FI_StartOfSDU,

                  v_RLC_Data1);

  //@siclog "Step 6" siclog@

  // Check: Does the UE transmit a RLC STATUS PDU?

  p_RLC_Rec.AM_VTA := 1;

  f_RxSTATUS_PDU_ACK(p_RLC_Rec);

  //@siclog "Step 7" siclog@

  // SS transmits AMD PDU#6 with SN=Receiving_AM_Window_Size+2

  p_RLC_Rec.AM_VTS := tsc_AM_Window_Size + 2;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 20, 20);

  f_TxAMD_OnePDU (p_RLC_Rec,

                  cs_TimingInfo_Now,

                  tsc_P_NoPoll,

                  tsc_FI_EndOfSDU,

                  v_RLC_Data1);

  //@siclog "Step 8" siclog@

  // Check: Does UE return RLC SDU#3 within 1s?

  p_RLC_Rec.AM_VRR := 0;

  t_WatchDog.start(1.0);

  alt {

    [] DRB.receive (car_DRB1_RLCAmDataInd (eutra_Cell1,

                                           cr_AMD_PDUoneSDU(p_RLC_Rec.AM_VRR, tsc_P_NoPoll, {p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3]} )))

      {

        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 8");

      }

    [] t_WatchDog.timeout

      {

        // UE does not return RLC SDU#3 within 1s

      }

  }; // alt

  //@siclog "Step 9" siclog@

  // SS transmits AMD PDU#6 with SN=1

  p_RLC_Rec.AM_VTS := 1;

  v_RLC_Data1 := substr (p_RLC_Rec.RLC_DataList[tsc_RLC_SDU3], 20, 20);

  f_TxAMD_OnePDU (p_RLC_Rec,

                  cs_TimingInfo_Now,

                  tsc_P_NoPoll,

                  tsc_FI_EndOfSDU,

                  v_RLC_Data1);

  //@siclog "Step 10" siclog@

  // Check: Does UE return RLC SDU#3 with its first AMD PDU set to SN=1?

  p_RLC_Rec.AM_VRR := 1;

  DRB.receive (car_DRB1_RLCAmDataInd (eutra_Cell1,

                                      cr_AMD_PDUoneSDU(p_RLC_Rec.AM_VRR,

                                                       tsc_P_Poll,

                                                       {v_SDU3_Received}))) ;

  p_RLC_Rec.AM_VRR := (p_RLC_Rec.AM_VRR + 1) mod 1024;

  f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 10");

  //@siclog "Step 10A" siclog@

  // Send STATUS PDU upon Pollbit.

  f_TxSTATUS_PDU(p_RLC_Rec);

} // fl_EUTRA_7_2_3_21_TestBody


New Functions: 

	  function f_EUTRA_508RRC_IntraLTE_HO_IntraCell_72321 ( CellId_Type                           p_CellId,

                                                  template (value) TimingInfo_Type      p_TimingInfo    := cs_TimingInfo_Now,

                                                  template (value) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_Def,

                                                  C_RNTI                                p_C_RNTI        := tsc_C_RNTI_Def2,

                                                  template (omit)  MeasConfig           p_MeasConfig    := omit)  runs on EUTRA_PTC

  { /* @sic R5s100491: enhancement to support MeasConfig sic@ */

    /* @sic R5s100524: common change of HO functions sic@ */

    var DRB_IdentityList_Type v_DRB_IdentityList;

    var integer k;

    var SubFrameTiming_Type v_Timing;

    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();

    var NextHopChainingCount v_NextHopChainingCount := 0; //According to 36.508 cl. 4.6.1 RRCConnectionReestablishment

    var Frequency_IE_Type v_Frequency_IE := f_EUTRA_CellInfo_GetFrequencyIEs (p_CellId);

    var CarrierFreqEUTRA v_CarrierFreqEutra := v_Frequency_IE.UL_DL_Earfcn;

    var template (value) MobilityControlInfo v_MobilityControlInfo := f_Generate_cs_MobilityControlInfo_HO_CRNTI(p_CellId, v_CarrierFreqEutra, p_C_RNTI);

    var template (value) RadioResourceConfigDedicated v_RadioResourceConfigDedicated := f_Generate_cs_508_RadioResourceConfigDedicated_HO (p_CellId);

    //Build DRB identity list from inputed radiobearer config list

    for (k:=0; k < lengthof(p_DrbConfigList); k:= k+1) {

      v_DRB_IdentityList[k] := valueof ( p_DrbConfigList[k].Id.Drb );

    }

    /* for intra cell HO the HO procedure and the SS configuration must be executed with timing info.

     * This is to make sure SS reconfiguration is done properly and in time before UE access the new cell.

     * The timing info used is based on the information provided in TTCN CR R5s100178 */

    if ( ischosen ( p_TimingInfo.SubFrame ) )

      {

        //Use the provided timing iformation

        v_Timing := valueof(p_TimingInfo.SubFrame);

      }

    else //i.e. p_TimingInfo is "now" (either not provided or provided as "now")

      {

        //Get next sending occassion

        v_Timing := f_EUTRA_GetNextSendOccasion ( p_CellId, 300); // R5-106675

      }

    //Transmit RRCConnectionReconfiguration/HO command to UE

    f_EUTRA_RRC_ConnReconfigHO_Common(p_CellId,

                                      p_MeasConfig,

                                      v_MobilityControlInfo,

                                      v_RadioResourceConfigDedicated,

                                      cs_508SecurityConfigHO_IntraLTE,

                                      cs_TimingInfo (v_Timing.SFN.Number, v_Timing.Subframe.Number));

    /* Note: current implementation releases SRBs and reconfigures C-RNTI 5ms after sending the RRCConnectionReconfiguration

       => RLC ACK is not taken into consideration */

    v_Timing := f_EUTRA_TimingInfoAdd(v_Timing, 5);   // add 5ms

    //Release SRBs and DRBs

    f_EUTRA_SS_SRBs_DRBs_Release ( p_CellId,

                                   cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                   v_DRB_IdentityList );

    //Configure C-RNTI based contention resolution in target cell 1 and assign new C-RNTI to the cell

    f_EUTRA_SS_ConfigRachProcedure_HO(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number), p_C_RNTI);

    //Save new C-RNTI in cell configuration for target cell (which is also source cell)

    f_EUTRA_CellInfo_SetC_RNTI ( p_CellId, p_C_RNTI );

    //Setup SRBs and DRBs again

    v_Timing := f_EUTRA_TimingInfoAdd(v_Timing, 5);   // add 5ms

    f_EUTRA_SS_SRBs_DRBs_Config ( p_CellId,

                                  cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                  p_DrbConfigList );

    //Enable RRC Integrity and Ciphering at SS target cell

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest ( p_CellId,

                                                               v_Auth_Params,

                                                               v_NextHopChainingCount,

                                                               cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),

                                                               v_DRB_IdentityList );

    f_EUTRA_Security_Set ( v_Auth_Params );

  }

